


# System shells



## BAT Advanced - 5



# FOR and VARIABLES

## ◆ Fill **adv1.bat**

```
@echo off
setlocal EnableDelayedExpansion
set count=0
for %%i in (1 2 3) do (
    set /a count+=1
    set /a var=%%i
    echo loop -%%i- : count is %count%, var is %var%
    echo    delayed: count is !count!, var is !var!
    echo.
)
echo Outside loop: count is %count%, var is %var%
endlocal
```

Use ! In place of %

unchanged for  
Set and %%

## ◆ Variable **EnabledDelayedExpansion**

- Variables evaluation inside <command> of **FOR**
- Modify the behavior of the CMD interpreter (*second pass*)
- Code must be **!VAR!** in place of **%VAR%**

# IF and VARIABLES

## ◆ Fill `adv2.bat`

```
@echo off
setlocal enabledelayedexpansion
set /a num1=%1
set /a num2=%2
```

```
if %num1% LEQ %num2% (
    set /a offset=!num2!-!num1!
) else (
    set /a offset=!num1!-!num2!
)
echo offset is %offset%
```

Use ! In place of %

## ◆ Variable `EnabledDelayedExpansion`

- Variables evaluation inside <command> of **IF**
- Modify the behavior of the CMD interpreter (*second pass*)
- Code must be **!VAR!** in place of **%VAR%**

# CALL and FUNCTIONS

## ◆ Fill `adv3.bat`

```
@echo off
setlocal enabledelayedexpansion
set /p mystr=please type a string of characters ...
call :strlen "%mystr%"
echo String "%mystr%" length is %ERRORLEVEL%
goto exit /b 0
```

```
:: ===== function STRLEN =====
:strlen
set "str=%~1"
for /l %%A in (0 1 4096) do (
    if "!str:~%%A,1!"==" exit /b %%A
)
exit /b 0
```

Spaces ??

Return as ERRORLEVEL

Empty string

## ◆ Use `exit /b XX` as function *return(XX)*

- The return value is in `%ERRORLEVEL%` after the **CALL**

# Search & Find



# Simple FIND

## ◆ Fill **adv4.bat**

```
@echo off
Echo foo > text.txt
Echo Bar >> text.txt
Echo FooBar >> text.txt
FIND /i "Bar" text.txt
Echo -----
FIND /V /n "Foo" text.txt
Echo -----
type text.txt | find /i /n "foo"
Del /Q text.txt
```

Will not display file name


## ◆ Command **FIND** has some arguments

- **/v** will display lines without match
- **/n** will add line numbers
- Can be used as a filter with « pipe »

# FIND inside FOR

## ◆ Fill **adv4.bat**

```
:: How many lines with EasterEgg in this script ?
@echo off
set ff=%~f0
set /a eggs=0
for /f "usebackq delims=" %%A in (`type %%ff%% ^| ^
FIND /i "easteregg"`) do (
    set /a eggs+=1
)
echo %eggs% lines containing easterEgg in %~nx0
```



## ◆ Note multiple escapes

- **usebackq** because **find** string "..."
- **%%ff%%** because in a quoted section
- **^** at end of line to continue command on next line



# Command FINDSTR

## ◆ Fill adv4.bat

```
@echo off
Echo foo > text.txt
Echo Bar >> text.txt
Echo FooBar >> text.txt
type text.txt | findstr ^[fF]
Del /Q text.txt
```

Lines starting with f or F

## ◆ FINDSTR accepts metacharacters

- . One of Any character
- \* Zero or any number of any character
- ^ Start of line
- \$ End of line
- [class] Any character in the group i.e [Aez12]
- [^class] Any character NOT in the group [^ezEZ]
- [x-y] Any character in the interval [0-9]
- \x escape for literal usage of metachar 'x'
- \<xyz « xyz » at start of word
- xyz\> « xyz » at end of word

# Command WHERE

## ◆ Fill adv5.bat

```
@echo off  
  
set /p file_name=Enter pattern for file name:  
  
WHERE /T /R "%LOCALAPPDATA%" "%file_name%"
```

## ◆ WHERE accepts some metacharacters

- ? One of Any character
- \* Zero or any number of any character

# Redirections



# OUT and IN flows

## ◆ Fill flow1.bat

```
@echo off
>text.txt echo "FOO"
echo bar >>text.txt
1>>text.txt echo "ACME"
md FOO/BAR 2>log.txt
2>>log.txt md FOO/BAR

echo ----more---- & more text.txt
echo ---- ^< ---- & <text.txt more
echo ----0^< ---- & 0<text.txt more
echo ----FIND---- & <text.txt FIND "ar"
```

## ◆ Standard flows

- Standard IN is flow 0 (implicit on < or <<)
- Standard OUT is flow 1 (implicit on > or >>)
- ERROR OUT is flow 2

# User defined flows

## ◆ Fill flow2.bat

```
@echo off
1>test.txt echo first line
1>>test.txt echo second line

<&3>LogInput.txt type test.txt
type LogInput.txt
echo.

0<&3>>logInput.txt type test.txt
type LogInput.txt

del /Q LogInput.txt
del /Q test.txt
```

## ◆ User defined flows

- Use an explicit number i.e **flow 3**
- Use redirection operators **<&3** or **&>3**

# Compare



# Compare files with FC

## ◆ Fill adv6.bat

```
@echo off

echo -----BEFORE----- > tempo.txt
Type %~nx0 >> tempo.txt
echo -----AFTER----- >> tempo.txt

FC tempo.txt %~nx0
del /Q tempo.txt
```

## ◆ FC will give differences between the two files

- Works on files with different sizes
- May receive sets of files
- /L or /U for ASCII or UNICODE text files
- /C to ignore case
- /B for BINARY files
- /LBn to stop after n differences

# Compare files with COMP

## ◆ Fill adv7.bat

```
@echo off

echo -----BEFORE----- > temp1.txt
Type %~nx0 >> temp1.txt
echo -----AFTER----- > temp2.txt
Type %~nx0 >> temp2.txt

COMP temp1.txt temp2.txt /A /M
del /Q temp*.txt
```

## ◆ COMP will give differences between the two files

- Give details if files have the same
- May receive sets of files
- /A to give the result in ASCII
- /D to give the result in decimal
- /C to ignore ASCII case (A = a)



# Copy/Compare with XCOPY

## ◆ Fill adv8.bat

```
@echo off

Set DIR=%USERPROFILE%\LABS
Set BKP=%USERPROFILE%\BACKUP\LABS
Echo .txt > nsav
Echo .tmp >> nsav

xcopy %DIR%\*.* %BKP%\*.* /D /E /H /I /EXCLUDE:nsav /Y
del /Q nsav
```

## ◆ XCOPY will update the destination with any new thing

- /D updates files that have been modified
- /E iterates copy through folder tree
- /H copy hidden files
- /I Creates destination folder if necessary
- /EXCLUDE ignore files with a match in nsav (\*.txt \*.tmp)
- /Y copy without user confirmation

# Control



# Applications Life & Death

## ◆ Fill adv9.bat

```
@echo off
START notepad.exe

TASKLIST /FI "IMAGENAME eq notepad.exe"
Set /P answ=Enter PID of NOTEPAD:

TASKKILL /PID %answ%
```

## ◆ **TASKLIST** has many options

- Modules (dll)
- Services
- Output (CSV etc.)

# What is my IP adress ?

- ◆ `IPCONFIG /all`

- ◆ `IPCONFIG | FINDSTR "IPv4"`